D.2. Linear Algebra in Maple

D.J. Jeffrey and Robert M. Corless
Department of Applied Mathematics
and the Ontario Research Centre for Computer Algebra,
The University of Western Ontario, London, Ontario, Canada N6A 5B7

1 Introduction

Maple is a general purpose computational system which combines symbolic computation with exact and approximate (floating-point) numerical computation, and which as well offers a comprehensive suite of scientific graphics. The main library of functions is written in the Maple programming language, a rich language designed to allow easy access to advanced mathematical algorithms. A special feature of Maple is user access to the source code for the library, including the ability to trace Maple's execution and see its internal workings; only the parts of Maple that are compiled, for example the kernel, cannot be traced. Another feature is that users can link to LAPACK library routines transparently, and thereby benefit from fast and reliable floating-point computation. The development of Maple started in the early 80s, and the company **Maplesoft** was founded in 1988. A strategic partnership with NAG Inc. in 2000 brought highly efficient numerical routines to Maple including LAPACK.

There are two linear algebra packages in Maple: `LinearAlgebra` and `linalg`. The `linalg` package is older, and is considered obsolete; it was replaced by `LinearAlgebra` in Maple 6. Here we describe only the `LinearAlgebra` package. The reader should be careful when reading other reference books, or the Maple help pages, to check whether reference is made to `vector, matrix, array` (notice the lower-case initial letter), which means that the older package is being discussed, or to `Vector, Matrix, Array` (with an upper-case initial letter), which means that the newer package is being discussed.

Facts

1. Maple commands are typed after a prompt symbol, which by default is 'greater than' (>). In examples below, keyboard input is simulated by prefixing the actual command typed with the prompt symbol.

2. In the examples below, some of the commands are too long to fit on one line. In such cases, the MAPLE continuation character backslash (\ ) is used to break the command across a line.

3. MAPLE commands are terminated by either semicolon (;) or colon (:). Before MAPLE 10, a terminator was required, but in the MAPLE 10 GUI it can be replaced by a carriage return. The semicolon terminator allows the output of a command to be displayed, while the colon suppresses the display (but the command still executes).

4. To access the commands described below, load the `LinearAlgebra` package by typing the command (after the prompt, as shown)

   `> with( LinearAlgebra );`

   If the package is not loaded, then either a typed command will not be recognized, or a different command with the same name will be used.

5. The results of a command can be assigned to one or more variables. Thus

   `> a := 1 ;`

   assigns the value 1 to the variable a, while

   `> (a,b,c) := 1,2,3 ;`

   assigns a the value 1, b the value 2 and c the value 3.

   CAUTION: The operator colon-equals (:=) is assignment, while the operator equals (=) defines an equation with a left-hand side and a right-hand side.

6. A sequence of expressions separated by commas is an expression sequence in MAPLE, and some commands return expression sequences, which can be assigned as above.

7. Ranges in MAPLE are generally defined using a pair of periods (..). The rules for the ranges of subscripts are given below.

2 Vectors

Facts:

1. In MAPLE, vectors are not just lists of elements. MAPLE separates the idea of the mathematical object `Vector` from the data object `Array` (see Subsection 4).

2. A MAPLE `Vector` can be converted to an `Array`, and an `Array` of appropriate shape can be converted to a `Vector`, but they cannot be used interchangeably in commands. See the help file for `convert` to find out about other conversions.

3. MAPLE distinguishes between column vectors, the default, and row vectors. The two types of vectors behave differently, and are not merely presentational alternatives.

Commands:

1. Generation of vectors.

   - `Vector( [`$x_1, x_2, \ldots$`] )` Construct a column vector by listing its elements. The length of the list specifies the dimension.
   - `Vector[column]( [`$x_1, x_2, \ldots$ `] )` Explicitly declare the column attribute.
   - `Vector[row]( [`$x_1, x_2, \ldots$ `] )` Construct a row vector by initializing its elements from a list.
   - `<` $v_1, v_2, \ldots$ `>` Construct a column vector with elements v_1, v_2, etc. An element can be another column vector.
   - `<` $v_1 | v_2 | \ldots$ `>` Construct a row vector with elements v_1, v_2, etc. An element can be another row vector. A useful mnemonic is that the vertical bars remind us of the column dividers in a table.
   - `Vector( n, k->f(k) )`. Construct an n-dimensional vector using a function $f(k)$ to define the elements. $f(k)$ is evaluated sequentially for k from 1 to n. The notation `k->f(k)` is MAPLE syntax for a univariate function.
   - `Vector( n, fill=v )` An n-dimensional vector with every element $\mathbf{v}$.
   - `Vector( n, symbol=v )` An n-dimensional vector containing symbolic components v_k.
   - `map( x->f(x), V )` Construct a new vector by applying function $f(x)$ to each element of the vector named `V`. CAUTION: the command is `map` *not* `Map`.

2. Operations and functions

   - `v[i]` Element i of vector $\mathbf{v}$. The result is a scalar.

     CAUTION: A symbolic reference `v[i]` is typeset as v_i on output in a Maple worksheet.
   - `v[p..q]` Vector consisting of elements v_i, $p \leq i \leq q$. The result is a `Vector`, even for the case `v[p..p]`. Either p or q can be negative, meaning the location is found by counting backwards from the end of the vector, with -1 being the last element.
   - `u+v, u-v` Add or subtract `Vectors` $\mathbf{u}$, $\mathbf{v}$.
   - `a*v` Multiply vector $\mathbf{v}$ by scalar a. Notice the operator is 'asterisk' ($*$).

- `u . v`, `DotProduct( u, v )` The inner product of `Vectors` **u** and **v**. See examples for complex conjugation rules. Notice the operator is 'period' (.) not 'asterisk' ($*$), because inner product is not commutative over the field of complex numbers.
- `Transpose( v ), v^%T` Change a column vector into a row vector, or *vice versa*. Complex elements are not conjugated.
- `HermitianTranspose( v ), v^%H` Transpose with complex conjugation.
- `OuterProductMatrix( u, v )` The outer product of `Vectors` **u** and **v** (ignoring the row/column attribute).
- `CrossProduct( u, v ), u &x v` The vector product, or cross product, of 3-dimensional vectors **u**, **v**.
- `Norm( v, 2 )` The 2-norm or Euclidean norm of vector **v**. Notice that the second argument, i.e. 2, is necessary, because `Norm( v )` defaults to the infinity norm, which is different from the default in many textbooks and software packages.
- `Norm( v, p )` The p-norm of **v**, namely $\left(\sum_{i=1}^{n} |v_i|^p\right)^{(1/p)}$.

Examples: In this subsection, the imaginary unit is the MAPLE default I. That is $\sqrt{-1} = I$. In the matrix section, we show how this can be changed. To save space, we shall mostly use row vectors in the examples.

1. Generate vectors. The same vector created different ways.

```
> Vector[row]([0,3,8]):  <0|3|8>:  Transpose(<0,3,8>):  Vector[row](3,i->i^2-1);
```

$$[0, 3, 8]$$

2. Selecting elements.

```
> V:=<a|b|c|d|e|f>:  V1 := V[2 .. 4]; V2:= V[-4 .. -1]; V3:= V[-4 .. 4];
```

$$V1 := [b, c, d]\ ,\quad V2 := [c, d, e, f]\ ,\quad V3 := [c, d]$$

3. A Gram–Schmidt exercise.

```
> u1 := <3|0|4>:  u2 := <2|1|1>:  w1n := u1/Norm( u1, 2 );
```

$$w1n := [3/5, 0, 4/5]$$

```
> w2 := u2 - (u2 . w1n)*w1n; w2n := w2/Norm( w2, 2 );
```

$$w2n := \left[\frac{2\sqrt{2}}{5},\ \frac{\sqrt{2}}{2},\ -\frac{3\sqrt{2}}{10}\right]$$

4. Vectors with complex elements. Define column vectors $\mathbf{u_c}$,$\mathbf{v_c}$ and row vectors $\mathbf{u_r}$,$\mathbf{v_r}$.

   `> uc := <1+I,2>:  ur := Transpose( uc ):  vc := <5,2-3*I>:  vr := Transpose( vc ):`

   The inner product of column vectors conjugates the first vector in the product, and the inner product of row vectors conjugates the second.

   `> inner1 := uc . vc; inner2 := ur . vr;`

   `inner1 := 9-11 I  , inner2 := 9+11 I`

   MAPLE computes the product of two similar vectors, i.e., both rows or both columns, as a true mathematical inner product, since that is the definition possible; in contrast, if the user mixes row and column vectors, then MAPLE does not conjugate:

   `> but := ur . vc;`

   $$but := 9 - I$$

CAUTION: The use of period (.) with complex row and column vectors together differs from the use of period (.) with complex $1 \times m$ and $m \times 1$ matrices. In case of doubt, use matrices and conjugate explicitly where desired.

3 Matrices

Facts:

1. One-column matrices and vectors are not interchangeable in MAPLE.

2. Matrices and 2-dimensional arrays are not interchangeable in MAPLE.

Commands:

1. Generation of Matrices.

   - `Matrix( [[`$a, b, \ldots$`],[`$c, d, \ldots$`],...] )` Construct a matrix row-by-row, using a list of lists.
   - `<< a|b|...>,<c|d|...>,...>` Construct a matrix row-by-row using vectors. Notice that the rows are specified by row vectors, requiring the | notation.
   - `<< a,b,...>|< c,d,... >|...>` Construct a matrix column-by-column using vectors. Notice that each vector is a column, and the columns are joined using | , the column operator.

     CAUTION: Both variants of the `<< ... >>` constructor are meant for interactive use, not programmatic use. They are slow, especially for large matrices.

- `Matrix( n, m, (i,j)->f(i,j) )`   Construct a matrix $n \times m$ using a function $f(i, j)$ to define the elements. $f(i, j)$ is evaluated sequentially for i from 1 to n and j from 1 to m. The notation `(i,j)->f(i,j)` is MAPLE syntax for a bivariate function $f(i, j)$.
- `Matrix( n, m, fill=a )`   An $n \times m$ matrix with each element equal to a.
- `Matrix( n, m, symbol=a )`   An $n \times m$ matrix containing subscripted entries a_{ij}.
- `map( x->f(x), M )`   A matrix obtained by applying $f(x)$ to each element of M.

  [CAUTION: the command is `map` *not* `Map`.]
- `< < A | B>, < C | D> >`   Construct a partitioned or block matrix from matrices A, B, C, D. Note that `< A|B >` will be formed by adjoining columns; the block `< C|D >` will be placed below `< A|B >`. The MAPLE syntax is similar to a common textbook notation for partitioned matrices.

2. Operations and functions

- `M[i,j]`   Element i, j of matrix M. The result is a scalar.
- `M[1..-1,k]`   Column k of `Matrix` M. The result is a `Vector`.
- `M[k,1..-1]`   Row k of `Matrix` M. The result is a row `Vector`.
- `M[p..q,r..s]`   Matrix consisting of submatrix m_{ij}, $p \le i \le q$, $r \le j \le s$. In HANDBOOK notation, $M[\{p, \dots, q\}, \{r, \dots, s\}]$.
- `Transpose( M )`, `M∧%T`   Transpose matrix M, without taking the complex conjugate of the elements.
- `HermitianTranspose( M )`, `M∧%H`   Transpose matrix M, taking the complex conjugate of elements.
- `A ± B`   Add/subtract compatible matrices or vectors A, B.
- `A.B`   Product of compatible matrices or vectors A, B. The examples below detail the ways in which MAPLE interprets products, since there are differences between MAPLE and other software packages.
- `MatrixInverse( A )`, `A∧(−1)`   Inverse of matrix A.
- `Determinant( A )`   Determinant of matrix A.
- `Norm( A, 2 )`   The (subordinate) 2-norm of matrix A, namely $\max_{\|u\|_2=1} \|Au\|_2$ where the norm in the definition is the vector 2-norm.

CAUTIONS:

(a) Notice that the second argument, i.e. 2, is necessary, because `Norm( A )` defaults to the infinity norm, which is different from the default in many textbooks and software packages.

(b) Notice also that this is the largest *singular value* of A, and is usually different than the Frobenius norm $\|A\|_F$, accessed by `Norm( A, Frobenius )`, which is the Euclidean norm of the vector of elements of the matrix A.

(c) Unless A has floating point entries, this norm will not usually be computable explicitly, and it may be expensive even to try.

- `Norm( A, p )` The (subordinate) matrix p-norm of A, for integers $p >= 1$ or $p =$ the symbol `infinity`.

Examples:

1. A matrix product.

```
> A := <<1|-2|3>,<0|1|1>>; B := Matrix( 3, 2, symbol=b ); C := A . B;
```

$$A := \begin{bmatrix} 1 & -2 & 3 \\ 0 & 1 & 1 \end{bmatrix}, \qquad B := \begin{bmatrix} b_{11} & b_{12} \\ b_{21} & b_{22} \\ b_{31} & b_{32} \end{bmatrix},$$

$$C := \begin{bmatrix} b_{11} - 2b_{21} + 3b_{31} & b_{12} - 2b_{22} + 3b_{32} \\ b_{21} + b_{31} & b_{22} + b_{32} \end{bmatrix}.$$

2. A Gram–Schmidt calculation re-visited.

If u_1, u_2 are $m \times 1$ column matrices, then the Gram–Schmidt process is often written in textbooks as

$$w_2 = u_2 - \frac{u_2^T u_1}{u_1^T u_1} u_1 \ .$$

Notice, however, that $u_2^T u_1$ and $u_1^T u_1$ are strictly 1×1 matrices. Textbooks often skip over the conversion of $u_2^T u_1$ from a 1×1 matrix to a scalar. MAPLE, in contrast, does not convert automatically. Transcribing the printed formula into MAPLE will cause an error. Here is the way to do it, reusing the earlier numerical data.

```
> u1 := <<3,0,4>>; u2 := <<2,1,1>>; r := u2^%T . u1; s := u1^%T . u1;
```

$$u1 := \begin{bmatrix} 3 \\ 0 \\ 4 \end{bmatrix}, \quad u2 := \begin{bmatrix} 2 \\ 1 \\ 1 \end{bmatrix}, \quad r := [10], \quad s := [25]$$

Notice the brackets in the values of `r` and `s`, because they are matrices. Since `r[1,1]` and `s[1,1]` are scalars, we write

```
> w2 := u2 - r[1,1]/s[1,1]*u1;
```

and re-obtain the result from the vector subsection. Alternatively, $u1$ and $u2$ can be converted to `Vectors` first and then used to form a proper scalar inner product.

```
> r := u2[1..-1,1] . u1[1..-1,1]; s := u1[1..-1,1] . u1[1..-1,1]; w2 := u2-r/s*u1;
```

$$r := 10\ , \quad s := 25\ , \quad w2 := \begin{bmatrix} 4/5 \\ 1 \\ -3/5 \end{bmatrix}$$

3. Vector-Matrix and Matrix-Vector products.

   Many textbooks equate a column vector and a one-column matrix, but this is not generally so in MAPLE. Thus

   ```
   > b := <1,2>; B := <<1,2>>; C := <<4|5|6>>;
   ```

   $$b := \begin{bmatrix} 1 \\ 2 \end{bmatrix} B := \begin{bmatrix} 1 \\ 2 \end{bmatrix} C := \begin{bmatrix} 4 & 5 & 6 \end{bmatrix}$$

   Only the product `B . C` is defined, and the product `b . C` causes an error.

   ```
   > B . C
   ```

   $$\begin{bmatrix} 4 & 5 & 6 \\ 8 & 10 & 12 \end{bmatrix}$$

   The rules for mixed products are

   ```
   Vector[row]( n ) . Matrix( n, m )          =  Vector[row]( m )
   Matrix( n, m ) . Vector[column]( m )       =  Vector[column]( n )
   ```

   The combinations `Vector( n ) . Matrix( 1, m )` and `Matrix( m, 1 ) . Vector[row]( n )` cause errors. If users do not want this level of rigour, then the easiest thing to do is to use only the `Matrix` declaration.

4. Working with matrices containing complex elements.

   First, notation: in linear algebra, I is commonly used for the identity matrix. This corresponds to the `eye` function in Matlab. However, by default, MAPLE uses I for the imaginary unit, as seen in Subsection 2. We can, however, use I for an identity matrix by changing the imaginary unit to something else, say `_i`.

```
> interface( imaginaryunit=_i ):
```

As the saying goes: an `_i` for an I and an I for an `eye`.

Now we can calculate eigenvalues using notation similar to introductory textbooks.

```
> A := <<1,2>|<-2,1>>; I := IdentityMatrix( 2 ); p := Determinant( x*I-A );
```

$$A := \begin{bmatrix} 1 & -2 \\ 2 & 1 \end{bmatrix}, \quad I := \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}, \quad p := x^2 - 2\,x + 5$$

Solving $p = 0$ we obtain eigenvalues $1+2i, 1-2i$. With the above setting of `imaginaryunit`, Maple will print these values as `1+2 _i, 1-2 _i`, but we have translated back to standard mathematical i, where $i^2 = -1$.

5. Moore-Penrose inverse. Consider `> M := Matrix( 3, 2, [[1,1],[a,a^2],[a^2,a]] );`, a 3×2 matrix containing a symbolic parameter a. We compute its Moore-Penrose pseudoinverse and a proviso guaranteeing correctness by the command

   `> ( Mi, p ) := MatrixInverse( M, method=pseudo, output=[inverse,proviso] );` which assigns the 2×3 pseudoinverse to `Mi` and an expression which if nonzero guarantees that `Mi` is the correct (unique) Moore-Penrose pseudoinverse of M. Here we have

$$Mi := \begin{bmatrix} \left(2+2\,a^3+a^2+a^4\right)^{-1} & -\frac{a^3+a^2+1}{a(a^5+a^4-a^3-a^2+2\,a-2)} & \frac{a^4+a^3+1}{a(a^5+a^4-a^3-a^2+2\,a-2)} \\ \left(2+2\,a^3+a^2+a^4\right)^{-1} & \frac{a^4+a^3+1}{a(a^5+a^4-a^3-a^2+2\,a-2)} & -\frac{a^3+a^2+1}{a(a^5+a^4-a^3-a^2+2\,a-2)} \end{bmatrix}$$

   and $p = a^2 - a$. Thus if $a \neq 0$ and $a \neq 1$ the computed pseudoinverse is correct. By separate computations we find that the pseudoinverse of $M|_{a=0}$ is

$$\begin{bmatrix} 1/2 & 0 & 0 \\ 1/2 & 0 & 0 \end{bmatrix}$$

   and that the pseudoinverse of $M|_{a=1}$ is

$$\begin{bmatrix} 1/6 & 1/6 & 1/6 \\ 1/6 & 1/6 & 1/6 \end{bmatrix}$$

   and *moreover that these are not special cases of the generic answer returned previously*. In a certain sense this is obvious: the Moore-Penrose inverse is *discontinuous*, even for square matrices (consider $(A - \lambda I)^{-1}$, for example, as $\lambda \to$ an eigenvalue of A).

4 Arrays

Before describing MAPLE's `Array` structure, it is useful to say why MAPLE distinguishes between an `Array` and a `Vector` or `Matrix`, when other books and software systems do not. In linear algebra, two different types of operations are performed with vectors or matrices. The first type is described in Subsections 2 and 3, and comprises operations derived from the mathematical structure of vector spaces. The other type comprises operations that treat vectors or matrices as data arrays; they manipulate the individual elements directly. As an example, consider dividing the elements of `Array` $[1, 3, 5]$ by the elements of $[7, 11, 13]$ to obtain $[1/7, 3/11, 5/13]$.

The distinction between the operations can be made in two places: in the name of the operation or the name of the object. In other words we can overload the data objects or overload the operators. Systems such as MATLAB choose to leave the data object unchanged, and define separate operators. Thus, in MATLAB the statements $[1, 3, 5]/[7, 11, 13]$ and $[1, 3, 5]./[7, 11, 13]$ are different because of the operators. In contrast, MAPLE chooses to make the distinction in the data object, as will now be described.

Facts:

1. The MAPLE `Array` is a general data structure akin to arrays in other programming languages.
2. An array can have up to 63 indexes and each index can lie in any integer range.
3. The description here only addresses the overlap between MAPLE `Array` and `Vector`.

CAUTIONS:

1. A MAPLE `Array` might look the same as a vector or matrix when printed.

Commands:

1. Generation of arrays.
   - `Array(`$[x_1, x_2, \ldots]$`)` Construct an array by listing its elements.
   - `Array( m..n )` Declare an empty 1-dimensional array indexed from m to n.
   - `Array( v )` Use an existing `Vector` to generate an array.
   - `convert( v, Array )` Convert a `Vector v` into an `Array`. Similarly, a `Matrix` can be converted to an `Array`. See the help file for `rtable_options` for advanced methods to convert efficiently, in-place.
2. Operations (memory/stack limitations may restrict operations).

- `a ^ n` Raise each element of a to power n.
- `a * b`, `a + b`, `a - b` Multiply (add, subtract) elements of b by (to, from) elements of a.
- `a / b` Divide elements of a by elements of b. Division by zero will produce `undefined` or `infinity` (or exceptions can be caught by user-set traps, see the help file for `Numeric_Events`).

Examples:

1. Array arithmetic.

```
> simplify( (Array([25,9,4])*Array(1..3,x->x^2-1 )/Array( <5,3,2> ))^(1/2));
```

$$[0, 3, 4]$$

2. Getting `Vectors` and `Arrays` to do the same thing.

```
> Transpose( map( x->x*x, <1,2,3> ) ) -convert( Array( [1,2,3] )^2, Vector );
```

$$[0, 0, 0]$$

5 Equation solving and matrix factoring

Cautions:

1. If a matrix contains exact numerical entries, typically integers or rationals, then the material studied in introductory textbooks transfers to a computer algebra system without special considerations. However, if a matrix contains symbolic entries, then the fact that computations are completed without the user seeing the intermediate steps can lead to unexpected results.

2. Some of the most popular matrix functions are discontinuous when applied to matrices containing symbolic entries. Examples are given below.

3. Some algorithms taught to educate students about the concepts of linear algebra often turn out to be ill-advised in practice: computing the characteristic polynomial and then solving it to find eigenvalues, for example; using Gaussian elimination without pivoting on a matrix containing floating-point entries, for another.

Commands:

1. `LinearSolve( A, B )` The vector or matrix X satisfying $AX = B$.

2. `BackwardSubstitute( A, B ), ForwardSubstitute( A, B )` The vector or matrix X satisfying $AX = B$ when A is upper or lower triangular (echelon) form respectively.

3. `ReducedRowEchelonForm( A ).` The reduced row-echelon form (RREF) of the matrix A. For matrices with symbolic entries, see the examples below for recommended usage.

4. `Rank( A )` The rank of the matrix A.

   CAUTION: if A has floating-point entries, see the subection below on Numerical Linear Algebra. On the other hand, if A contains symbolic entries, then the rank may change discontinuously and the generic answer returned by `Rank` may be incorrect for some specializations of the parameters.

5. `NullSpace( A )` The nullspace (kernel) of the matrix A.

   CAUTION: if A has floating-point entries, see the subsection below on Numerical Linear Algebra. Again on the other hand, if A contains symbolic entries, the nullspace may change discontinuously and the generic answer returned by `NullSpace` may be incorrect for some specializations of the parameters.

6. `( P, L, U, R ) := LUDecomposition( A, method='RREF' )` The $PLUR$, or Turing, factors of the matrix A. See examples for usage.

7. `( P, L, U ) := LUDecomposition( A )` The PLU factors of a matrix A, when the RREF R is not needed. This is usually the case for a Turing factoring where R is guaranteed (or known a *priori*) to be I, the identity matrix, for all values of the parameters.

8. `( Q, R ) := QRDecomposition( A, fullspan )` The QR factors of the matrix A. The option `fullspan` ensures that Q is square.

9. `SingularValues( A )` See Subsection 8 Numerical Linear Algebra.

10. `ConditionNumber( A )` See Subsection 8 Numerical Linear Algebra.

Examples:

1. Need for Turing factoring.

   One of the strengths of Maple is computation with symbolic quantities. When standard linear algebra methods are applied to matrices containing symbolic entries, the user must be aware of new mathematical features which can arise. The main feature is the discontinuity of standard matrix functions, such as the reduced row-echelon form and the rank, both of which can be discontinuous. For example, the matrix

$$B = A - \lambda I = \begin{bmatrix} 7-\lambda & 4 \\ 6 & 2-\lambda \end{bmatrix}$$

   has the reduced row-echelon form

$$\text{ReducedRowEchelonForm}(B) = \begin{cases} \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix} & \lambda \neq -1, 10 \\[2ex] \begin{bmatrix} 1 & -4/3 \\ 0 & 0 \end{bmatrix} & \lambda = 10 \ , \\[2ex] \begin{bmatrix} 1 & 1/2 \\ 0 & 0 \end{bmatrix} & \lambda = -1 \ . \end{cases}$$

   Notice that the function is discontinuous precisely at the interesting values of λ. Computer algebra systems in general, and Maple in particular, return 'generic' results. Thus in Maple we have

```
> B := << 7-x | 4 >, < 6 | 2-x >>;
```

$$B := \begin{bmatrix} 7-x & 4 \\ 6 & 2-x \end{bmatrix}$$

```
> ReducedRowEchelonForm( B )
```

$$\begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}$$

   This difficulty is discussed at length in [CJ92, CJ97]. The recommended solution is to use Turing factoring (generalized PLU decomposition) to obtain the reduced row-echelon form with provisos. Thus, for example,

```
> A := <<1|-2|3|sin(x)>,<1|4*cos(x)|3|3*sin(x)>,<-1|3|cos(x)-3|cos(x)>>;
```

$$A := \begin{bmatrix} 1 & -2 & 3 & \sin x \\ 1 & 4\cos x & 3 & 3\sin x \\ -1 & 3 & \cos x - 3 & \cos x \end{bmatrix}$$

```
> ( P, L, U, R ) := LUDecomposition( A, method='RREF' ):
```

The generic reduced row-echelon form is then given by

$$R = \begin{bmatrix} 1 & 0 & 0 & (2\sin x\cos x - 3\sin x - 6\cos x - 3)/(2\cos x + 1) \\ 0 & 1 & 0 & \sin x/(2\cos x + 1) \\ 0 & 0 & 1 & (2\cos x + 1 + 2\sin x)/(2\cos x + 1) \end{bmatrix}$$

This shows a visible failure when $2\cos x + 1 = 0$, but the other discontinuity is invisible, and requires the U factor from the Turing ($PLUR$) factors,

$$U = \begin{bmatrix} 1 & -2 & 3 \\ 0 & 4\cos x + 2 & 0 \\ 0 & 0 & \cos x \end{bmatrix}$$

to see that the case $\cos x = 0$ also causes failure. In both cases (meaning the cases $2\cos x + 1 = 0$ and $\cos x = 0$), the RREF must be recomputed to obtain the singular cases correctly.

2. QR factoring.

   MAPLE does not offer column pivoting, so in pathological cases the factoring may not be unique, and will vary between software systems. For example,

   ```
   > A := <<0,0>|<5,12>>:  QRDecomposition( A, fullspan )
   ```

$$\begin{bmatrix} 5/13 & 12/13 \\ 12/13 & -5/13 \end{bmatrix}, \quad \begin{bmatrix} 0 & 13 \\ 0 & 0 \end{bmatrix}$$

6 Eigenvalues and eigenvectors

Facts:

1. In exact arithmetic, explicit expressions are not possible in general for the eigenvalues of a matrix of dimension 5 or higher.

2. When it has to, MAPLE represents polynomial roots (and hence eigenvalues) *implicitly* by the `RootOf` construct. Expressions containing `RootOf`s can be simplified and evaluated numerically.

Commands:

1. `Eigenvalues( A )` The eigenvalues of matrix A.

2. `Eigenvectors( A )` The eigenvalues and corresponding eigenvectors of A.

3. `CharacteristicPolynomial( A, 'x' )` The characteristic polynomial of A expressed using the variable x.

4. `JordanForm( A )` The Jordan form of the matrix A.

Examples:

1. Simple eigensystem computation.

   ```
   > Eigenvectors( <<7,6>|<4,2>> );
   ```

   $$\begin{bmatrix} -1 \\ 10 \end{bmatrix}, \begin{bmatrix} -1/2 & 4/3 \\ 1 & 1 \end{bmatrix}$$

   So the eigenvalues are -1 and 10 with the corresponding eigenvectors $[-1/2, 1]^T$ and $[4/3, 1]^T$.

2. A defective matrix.

   If the matrix is defective, then by convention the matrix of 'eigenvectors' returned by MAPLE contains one or more columns of zeros.

   ```
   > Eigenvectors( <<1,0>|<1,1>> );
   ```

   $$\begin{bmatrix} 1 \\ 1 \end{bmatrix}, \begin{bmatrix} 1 & 0 \\ 0 & 0 \end{bmatrix}$$

3. Larger systems.

   For larger matrices, the eigenvectors will use the MAPLE `RootOf` construction.

   $$A := \begin{bmatrix} 3 & 1 & 7 & 1 \\ 5 & 6 & -3 & 5 \\ 3 & -1 & -1 & 0 \\ -1 & 5 & 1 & 5 \end{bmatrix}$$

   `> ( L, V ) := Eigenvectors( A ):` The colon suppresses printing. The vector of eigenvalues is returned as

```
> L;
```

$$\begin{bmatrix} \texttt{RootOf}\left(\_Z^4 - 13\,\_Z^3 - 4\,\_Z^2 + 319\,\_Z - 386, \texttt{index} = 1\right) \\ \texttt{RootOf}\left(\_Z^4 - 13\,\_Z^3 - 4\,\_Z^2 + 319\,\_Z - 386, \texttt{index} = 2\right) \\ \texttt{RootOf}\left(\_Z^4 - 13\,\_Z^3 - 4\,\_Z^2 + 319\,\_Z - 386, \texttt{index} = 3\right) \\ \texttt{RootOf}\left(\_Z^4 - 13\,\_Z^3 - 4\,\_Z^2 + 319\,\_Z - 386, \texttt{index} = 4\right) \end{bmatrix}$$

This, of course, simply reflects the characteristic polynomial:

```
> CharacteristicPolynomial( A, 'x' );
```

$$x^4 - 13x^3 - 4x^2 + 319x - 386$$

The `Eigenvalues` command solves a 4th degree characteristic polynomial explicitly in terms of radicals unless the option `implicit` is used.

4. Jordan form.

   CAUTION: As with the reduced row-echelon form, the Jordan form of a matrix containing symbolic elements can be discontinuous. For example, given

$$A = \begin{bmatrix} 1 & t \\ 0 & 1 \end{bmatrix},$$

```
> ( J, Q ) := JordanForm( A, output=['J','Q'] );
```

$$J, Q := \begin{bmatrix} 1 & 1 \\ 0 & 1 \end{bmatrix}, \begin{bmatrix} t & 0 \\ 0 & 1 \end{bmatrix}$$

   with $A = QJQ^{-1}$. Note that Q is invertible precisely when $t \neq 0$. This gives a *proviso* on the correctness of the result: J will be the Jordan form of A only for $t \neq 0$, which we see is the generic case returned by MAPLE.

   CAUTION: Exact computation has its limitations, even without symbolic entries. If we ask for the Jordan form of the matrix

$$B = \begin{bmatrix} -1 & 4 & -1 & -14 & 20 & -8 \\ 4 & -5 & -63 & 203 & -217 & 78 \\ -1 & -63 & 403 & -893 & 834 & -280 \\ -14 & 203 & -893 & 1703 & -1469 & 470 \\ 20 & -217 & 834 & -1469 & 1204 & -372 \\ -8 & 78 & -280 & 470 & -372 & 112 \end{bmatrix},$$

a relatively modest 6×6 matrix with a triple eigenvalue 0, then the transformation matrix Q as produced by MAPLE has entries over 35,000 characters long. Some scheme of compression or large expression management is thereby mandated.

7 Linear algebra with modular arithmetic in Maple

There is a sub-package, `LinearAlgebra[Modular]`, designed for programmatic use, that offers access to modular arithmetic with matrices and vectors.

Facts:

1. The sub-package can be loaded by issuing the command `with( LinearAlgebra[Modular] );` which gives access to the commands

   ```
   [AddMultiple, Adjoint, BackwardSubstitute, Basis, CharacteristicPolynomial,
   ChineseRemainder, Copy, Create, Determinant, Fill, ForwardSubstitute, Identity, Inverse,
   LUApply, LUDecomposition, LinIntSolve, MatBasis, MatGcd, Mod, Multiply, Permute, Random,
   Rank, RankProfile, RowEchelonTransform, RowReduce, Swap, Transpose, ZigZag]
   ```

2. Arithmetic can be done modulo a prime p or in some cases a composite modulus m.

3. The relevant matrix and vector datatypes are `integer[4]`, `integer[8]`, `integer[]`, and `float[8]`. Use of the correct datatype can improve efficiency.

Examples: `> p := 13;`

```
> A := Mod( p, Matrix([[1,2,3],[4,5,6],[7,8,-9]]), integer[4] );
```

$$\begin{bmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \\ 7 & 8 & 4 \end{bmatrix}$$

```
> Mod( p, MatrixInverse( A ), integer[4] );
```

$$\begin{bmatrix} 12 & 8 & 5 \\ 0 & 11 & 3 \\ 5 & 3 & 5 \end{bmatrix}$$

CAUTIONS:

1. This is not to be confused with the `mod` utilities, which together with the inert `Inverse` command, can also be used to calculate inverses in a modular way.
2. One must always specify the datatype in `Modular` commands, or a cryptic error message will be generated.

8 Numerical linear algebra in Maple

The above subsections have covered the use of MAPLE for exact computations of the types met during a standard first course on linear algebra. However, in addition to exact computation, MAPLE offers a variety of floating-point numerical linear algebra support.

Facts:

1. MAPLE can compute with either 'hardware floats' or 'software floats',
2. A hardware float is IEEE double precision, with a mantissa of (approximately) 15 decimal digits.
3. A software float has a mantissa whose length is set by the MAPLE variable `Digits`.

CAUTIONS:

1. If an integer is typed with a decimal point, then MAPLE treats it as a software float.
2. Software floats are significantly slower that hardware floats, even for the same precision.

Commands:

1. `Matrix( n, m, datatype=float[8] )` An $n \times m$ matrix of hardware floats (initialization data not shown). The elements must be real numbers. The 8 refers to the number of bytes used to store the floating point real number.
2. `Matrix( n, m, datatype=complex(float[8] ))` An $n \times m$ matrix of hardware floats, including complex hardware floats. A complex hardware float takes two 8-byte storage locations.
3. `Matrix( n, m, datatype=sfloat )` An $n \times m$ matrix of software floats. The entries must be real and the precision is determined by the value of `Digits`.
4. `Matrix( n, m, datatype=complex(sfloat) )` As before with complex software floats.
5. `Matrix( n, m, shape=symmetric )` A matrix declared to be symmetric. MAPLE can take advantage of shape declarations such as this.

Examples:

1. A characteristic surprise.

   When asked to compute the characteristic polynomial of a floating-point matrix, MAPLE first computes eigenvalues (by a good numerical method) and then presents the characteristic polynomial in factored form, with good approximate roots. Thus

```
> CharacteristicPolynomial( Matrix( 2, 2, [[666,667],[665,666]],\
      datatype=float[8]), 'x' );
```

$$(x - 1331.99924924882612 - 0.0\,i)\,(x - 0.000750751173882235889 - 0.0\,i)$$

   Notice the signed zero in the imaginary part; though the roots in this case are real, approximate computation of the eigenvalues of a nonsymmetric matrix takes place over the complex numbers. *N.B.* the output above has been edited for clarity.

2. Symmetric matrices.

   If MAPLE knows that a matrix is symmetric, then it uses appropriate routines. Without the symmetric declaration, the calculation is

```
> Eigenvalues( Matrix( 2, 2, [[1,3],[3,4]], datatype=float[8]) );
```

$$\begin{bmatrix} -0.854101966249684707 + 0.0\,i \\ 5.85410196624968470 + 0.0\,i \end{bmatrix}$$

   With the declaration, the computation is

```
> Eigenvalues( Matrix( 2, 2, [[1,3],[3,4]], shape=symmetric, datatype=float[8]) );
```

$$\begin{bmatrix} -0.854101966249684818 \\ \\ 5.85410196624968470 \end{bmatrix}$$

   CAUTION: Use of the `shape=symmetric` declaration will force MAPLE to treat the matrix as being symmetric, even if it is not.

3. Printing of hardware floats.

   MAPLE prints hardware floating-point data as 18-digit numbers. This does not imply that all 18 digits are correct; MAPLE prints the hardware floats this way so that a cycle of converting from binary to decimal and back to binary will return to exactly the starting binary floating-point number. Notice in the previous example that the last 3 digits differ between the two function calls. In fact neither set of digits is correct, as a calculation in software floats with higher precision shows.

```
> Digits := 30: Eigenvalues( Matrix( 2, 2, [[1,3],[3,4]], \
     datatype=sfloat, shape=symmetric ) );
```

$$\begin{bmatrix} -0.854101966249684544613760503093 \\ 5.85410196624968454461376050310 \end{bmatrix}$$

4. NullSpace. Consider

   ```
   > B := Matrix( 2, 2, [[666,667],[665,666]] ):  A := Transpose(B).B;
   ```

   We make a floating-point version of A by

   ```
   > Af := Matrix( A, datatype=float[8] );
   ```

   and then take the `NullSpace` of both `A` and `Af`. The nullspace of `A` is correctly returned as the empty set—A is not singular (in fact its determinant is 1). The nullspace of `Af` is correctly returned as

   $$\left\{ \begin{bmatrix} -0.707637442412755612 \\ 0.706575721416702662 \end{bmatrix} \right\}$$

   The answers are different—quite different—even though the matrices differ only in datatype. The surprising thing is that both answers are *correct*: MAPLE is doing the right thing in each case. See [Cor93] for a detailed explanation, but note that `Af` times the vector in the reported nullspace is about $3.17 \cdot 10^{-13}$ times the 2-norm of `Af`.

5. Approximate Jordan form (not!)[1]

   As noted previously, the Jordan form is discontinuous as a function of the entries of the matrix. This means that rounding errors may cause the computed Jordan form of a matrix with floating-point entries to be incorrect, and for this reason MAPLE refuses to compute the Jordan form of such a matrix.

6. Conditioning of eigenvalues.

   To explore MAPLE's facilities for the conditioning of the unsymmetric matrix eigenproblem, consider the matrix 'gallery(3)' from MATLAB.

   ```
   > A := Matrix( [[-149,-50,-154], [537,180,546], [-27,-9,-25]] ):
   ```

   The Maple command `EigenConditionNumbers` computes estimates for the reciprocal condition numbers of each eigenvalue, and estimates for the reciprocal condition numbers of the computed eigenvectors as well. At this time, there are no built-in facilities for the computation of the sensitivity of arbitrary invariant subspaces.

[1] An out-of-date humorous reference.

```
> E,V,rconds,rvecconds := EigenConditionNumbers( A, output=['values',\
    'vectors','conditionvalues','conditionvectors'] ):
> seq( 1/rconds[i], i=1..3 );
                                        417.6482708, 349.7497543, 117.2018824
```

A separate computation using the definition of the condition number of an eigentriplet $(\mathbf{y}^*, \lambda, \mathbf{x})$ (see §3.5) as

$$C_\lambda = \frac{\|\mathbf{y}^*\|_\infty \|\mathbf{x}\|_\infty}{|\mathbf{y}^* \cdot \mathbf{x}|}$$

gives exact condition numbers (in the infinity norm) for the eigenvalues 1, 2, 3 as 399, 252, and 147. We see that the estimates produced by `EigenConditionNumbers` are of the right order of magnitude.

9 Canonical forms

There are several canonical forms in Maple: Jordan form, Smith form, Hermite form, and Frobenius form, to name a few. In this section we talk only about the Smith form (defined in §2.1.5 and §4.4.2).

Commands:

1. `SmithForm( B, output=['S','U','V'] )` Smith form of B.

Examples: The Smith form of

$$B = \begin{bmatrix} 0 & -4y^2 & 4y & -1 \\ -4y^2 & 4y & -1 & 0 \\ 4y & -1 & 4\left(2y^2-2\right)y & 4\left(y^2-1\right)^2 y^2 - 2y^2 + 2 \\ -1 & 0 & 4\left(y^2-1\right)^2 y^2 - 2y^2 + 2 & -4\left(y^2-1\right)^2 y \end{bmatrix}$$

is

$$S = \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1/4\left(2y^2-1\right)^2 & 0 \\ 0 & 0 & 0 & (1/64)\left(2y^2-1\right)^6 \end{bmatrix}$$

Maple also returns two unimodular (over the domain $\mathbb{Q}[y]$) matrices $\mathbf{u}$ and $\mathbf{v}$ for which $A = U.S.V$.

10 Structured matrices

Facts:

1. Computer algebra systems are particularly useful for computations with structured matrices.

2. User-defined structures may be programmed using *index functions*. See the help pages for details.

3. Examples of built-in structures include *symmetric*, *skew-symmetric*, *Hermitian*, *Vandermonde*, and *Circulant* matrices.

Examples: **Generalized Companion Matrices**. MAPLE can deal with several kinds of generalized companion matrices. A generalized companion matrix[2] *pencil* of a polynomial $p(x)$ is a pair of matrices C_0, C_1 such that $\det(xC_1 - C_0) = 0$ precisely when $p(x) = 0$. Usually, in fact, $\det(xC_1 - C_0) = p(x)$, though in some definitions proportionality is all that is needed. In the case $C_1 = I$, the identity matrix, we have $C_0 = C(p(x))$ is *the companion matrix* of $p(x)$. MATLAB's `roots` function computes roots of polynomials by first computing the eigenvalues of the companion matrix, a venerable procedure only recently proved stable.

The generalizations allow direct use of alternative polynomial bases, such as the Chebyshev polynomials, Lagrange polynomials, Bernstein (Bézier) polynomials, and many more. Further, the generalizations allow the construction of generalized companion matrix pencils for *matrix polynomials*, allowing one to easily solve *nonlinear eigenvalue problems*.

We give three examples below.

If $p := 3 + 2x + x^2$, then `CompanionMatrix( p, x )` produces 'the' (standard) companion matrix (also called *Frobenius form* companion matrix):

$$\begin{bmatrix} 0 & -3 \\ 1 & -2 \end{bmatrix}$$

and it is easy to see that $\det(tI - C) = p(t)$. If instead

$$p := B_0^3(x) + 2B_1^3(x) + 3B_2^3(x) + 4B_3^3(x)$$

where $B_k^n(x) = \binom{n}{k}(1-x)^{n-k}(x+1)^k$ is the kth Bernstein (Bézier) polynomial of degree n on the interval $-1 \le x \le 1$, then `CompanionMatrix( p, x )` produces the pencil (note that this is not in Frobenius form)

$$C_0 = \begin{bmatrix} -3/2 & 0 & -4 \\ 1/2 & -1/2 & -8 \\ 0 & 1/2 & -\frac{52}{3} \end{bmatrix}$$

[2]Sometimes known as 'colleague' or 'comrade' matrices, an unfortunate terminology that inhibits keyword search.

$$C_1 = \begin{bmatrix} 3/2 & 0 & -4 \\ 1/2 & 1/2 & -8 \\ 0 & 1/2 & -\frac{20}{3} \end{bmatrix}$$

(from a formula by Jonsson & Vavasis [JV05] and independently by J. Winkler [Win04]), and we have $p(x) = \det(xC_1 - C_0)$. Note that the program does *not* change the basis of the polynomial $p(x)$ of equation (10) to the monomial basis (it turns out that $p(x) = 20 + 12x$ in the monomial basis, in this case: note that C_1 is singular). It is well-known that changing polynomial bases can be ill-conditioned, and this is why the routine avoids making the change.

Next, if we choose nodes $[-1, -1/3, 1/3, 1]$ and look at the degree 3 polynomial taking the values $[1, -1, 1, -1]$ on these four nodes, then `CompanionMatrix( values, nodes )` gives C_0 and C_1 where C_1 is the 5×5 identity matrix with the $(5, 5)$ entry replaced by 0, and

$$C_0 = \begin{bmatrix} 1 & 0 & 0 & 0 & -1 \\ 0 & 1/3 & 0 & 0 & 1 \\ 0 & 0 & -1/3 & 0 & -1 \\ 0 & 0 & 0 & -1 & 1 \\ -\frac{9}{16} & \frac{27}{16} & -\frac{27}{16} & \frac{9}{16} & 0 \end{bmatrix}.$$

We have that $\det(tC_1 - C_0)$ is of degree 3 (in spite of these being 5×5 matrices), and that this polynomial takes on the desired values ± 1 at the nodes. Therefore the finite eigenvalues of this pencil are the roots of the given polynomial. See [CW04, Cor04], for example, for more information.

Finally, consider the nonlinear eigenvalue problem below: find the values of x such that the matrix C with $C_{ij} = T_0(x)/(i + j + 1) + T_1(x)/(i + j + 2) + T_2(x)/(i + j + 3)$ is singular. Here $T_k(x)$ means the kth Chebyshev polynomial, $T_k(x) = \cos(k \cos^{-1}(x))$. We issue the command

```
> ( C0, C1 ) := CompanionMatrix( C, x );
```

from which we find

$$C0 = \begin{bmatrix} 0 & 0 & 0 & -2/15 & -1/12 & -\frac{2}{35} \\ 0 & 0 & 0 & -1/12 & -\frac{2}{35} & -1/24 \\ 0 & 0 & 0 & -\frac{2}{35} & -1/24 & -\frac{2}{63} \\ 1 & 0 & 0 & -1/4 & -1/5 & -1/6 \\ 0 & 1 & 0 & -1/5 & -1/6 & -1/7 \\ 0 & 0 & 1 & -1/6 & -1/7 & -1/8 \end{bmatrix}$$

and

$$C1 = \begin{bmatrix} 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 2/5 & 1/3 & 2/7 \\ 0 & 0 & 0 & 1/3 & 2/7 & 1/4 \\ 0 & 0 & 0 & 2/7 & 1/4 & 2/9 \end{bmatrix}.$$

This uses a formula from [Goo61], extended to matrix polynomials. The six generalized eigenvalues of this pencil include, for example, one near to $-0.6854 + 1.909i$. Substituting this eigenvalue in for x in C yields a three by three matrix with ratio of smallest to largest singular values $\sigma_3/\sigma_1 \approx 1.7 \cdot 10^{-15}$. This is effectively singular, and thus we have found the solutions to the nonlinear eigenvalue problem. Again note that the generalized companion matrix is not in Frobenius standard form, and that this process works for a variety of bases, including the Lagrange basis.

Circulant matrices and Vandermonde matrices

```
> A := Matrix( 3, 3, shape=Circulant[a,b,c] );
```

$$\begin{bmatrix} a & b & c \\ c & a & b \\ b & c & a \end{bmatrix}$$

```
> F3 := Matrix( 3, 3, shape=Vandermonde[[1,exp(2*Pi*I/3),exp(4*Pi*I/3)]] );
```

$$\begin{bmatrix} 1 & 1 & 1 \\ 1 & -1/2 + 1/2\,i\sqrt{3} & \left(-1/2 + 1/2\,i\sqrt{3}\right)^2 \\ 1 & -1/2 - 1/2\,i\sqrt{3} & \left(-1/2 - 1/2\,i\sqrt{3}\right)^2 \end{bmatrix}$$

It is easy to see that the `F3` matrix diagonalizes the circulant matrix `A`.

Toeplitz and Hankel matrices These can be constructed by calling `ToeplitzMatrix` and `HankelMatrix`, or by direct use of the `shape` option of the `Matrix` constructor.

```
> T := ToeplitzMatrix( [a,b,c,d,e,f,g] );
> T := Matrix( 4, 4, shape=Toeplitz[false,Vector(7,[a,b,c,d,e,f,g])] );
```

both yield a matrix that looks like

$$\begin{bmatrix} d & c & b & a \\ e & d & c & b \\ f & e & d & c \\ g & f & e & d \end{bmatrix},$$

though in the second case only 7 storage locations are used, whereas 16 are used in the first. This economy may be useful for larger matrices. The shape constructor for Toeplitz also takes a Boolean argument, `true` meaning symmetric.

Both Hankel and Toeplitz matrices may be specified with an indexed symbol for the entries:

`> H := Matrix( 4, 4, shape=Hankel[a] );` yields

$$\begin{bmatrix} a_1 & a_2 & a_3 & a_4 \\ a_2 & a_3 & a_4 & a_5 \\ a_3 & a_4 & a_5 & a_6 \\ a_4 & a_5 & a_6 & a_7 \end{bmatrix}.$$

11 Functions of matrices

The exponential of the matrix A is computed in the `MatrixExponential` command of Maple by *polynomial interpolation* (see §3.1.1) of the exponential at each of the eigenvalues of A, including multiplicities. In an exact computation context, this method is not so 'dubious' [George Labahn, personal communication]. This approach is also used by the general `MatrixFunction` command.

Examples: `> A := Matrix( 3, 3, [[-7,-4,-3],[10,6,4],[6,3,3]] ):`

```
> MatrixExponential( A );
```

$$\begin{bmatrix} 6-7\,e^1 & 3-4\,e^1 & 2-3\,e^1 \\ 10\,e^1-6 & -3+6\,e^1 & -2+4\,e^1 \\ 6\,e^1-6 & -3+3\,e^1 & -2+3\,e^1 \end{bmatrix} \tag{1}$$

Now a square root: `> MatrixFunction( A, sqrt(x), x ):`

$$\begin{bmatrix} -6 & -7/2 & -5/2 \\ 8 & 5 & 3 \\ 6 & 3 & 3 \end{bmatrix} \tag{2}$$

Another matrix square root example, for a matrix close to one that has no square root:

```
> A := Matrix( 2, 2, [[epsilon^2, 1], [0, delta^2] ] ):
> S := MatrixFunction( A, sqrt(x), x ):
> simplify( S ) assuming positive;
```

$$\begin{bmatrix} \epsilon & \frac{1}{\epsilon+\delta} \\ 0 & \delta \end{bmatrix} \tag{3}$$

If ϵ and δ both approach zero, we see that the square root has an entry that approaches infinity. Calling `MatrixFunction` on the above matrix with $\epsilon = \delta = 0$ yields an error message, `Matrix function x^(1/2) is not defined for this Matrix`, which is correct.

Now for the matrix logarithm. `> Pascal := Matrix( 4, 4, (i,j)->binomial(j-1,i-1) );`

$$\begin{bmatrix} 1 & 1 & 1 & 1 \\ 0 & 1 & 2 & 3 \\ 0 & 0 & 1 & 3 \\ 0 & 0 & 0 & 1 \end{bmatrix} \tag{4}$$

```
> MatrixFunction( Pascal, log(x), x );
```

$$\begin{bmatrix} 0 & 1 & 0 & 0 \\ 0 & 0 & 2 & 0 \\ 0 & 0 & 0 & 3 \\ 0 & 0 & 0 & 0 \end{bmatrix} \tag{5}$$

Now a function not covered in §3.1, instead of re-doing the sine and cosine examples: `> A := Matrix( 2, 2, [[-1/5, 1], [0, -1/5]] ):`

```
> W := MatrixFunction( A, LambertW(-1,x), x );
```

$$W := \begin{bmatrix} \mathrm{LambertW}(-1,-1/5) & -5\,\frac{\mathrm{LambertW}(-1,-1/5)}{1+\mathrm{LambertW}(-1,-1/5)} \\ 0 & \mathrm{LambertW}(-1,-1/5) \end{bmatrix} \tag{6}$$

```
> evalf( W );
```

$$\begin{bmatrix} -2.542641358 & -8.241194055 \\ 0.0 & -2.542641358 \end{bmatrix} \tag{7}$$

That matrix satisfies $W\exp(W) = A$, and is a *primary matrix function*. See [CGH+96] for more details about the Lambert W function.

Now the matrix sign function (cf. Section 3.1.6). Consider

```
> Pascal2 := Matrix( 4, 4, (i,j)->(-1)^(i-1)*binomial(j-1,i-1) );
```

$$\begin{bmatrix} 1 & 1 & 1 & 1 \\ 0 & -1 & -2 & -3 \\ 0 & 0 & 1 & 3 \\ 0 & 0 & 0 & -1 \end{bmatrix}. \tag{8}$$

Then we compute the matrix sign function of this matrix by `> S := MatrixFunction( Pascal2, csgn(z), z ):` which just turns out to be the matrix we started with (`Pascal2`).

Note: The complex 'sign' function we use here is not the usual complex sign function for scalars

$$\mathrm{signum}(re^{i\theta}) := e^{i\theta}$$

but rather (as desired for the definition of the matrix sign function)

$$\mathrm{csgn}(z) = \begin{cases} 1 & \text{if } \mathrm{Re}(z) > 0 \\ -1 & \text{if } \mathrm{Re}(z) < 0 \\ \mathrm{signum}(\mathrm{Im}(z)) & \text{if } \mathrm{Re}(z) = 0 \end{cases}$$

This has the side effect of making the function defined even when the input matrix has purely imaginary eigenvalues. The signum and `csgn` of 0 are both 0, by default, but can be specified differently if desired. CAUTION: Further, it is not the `sign` function in MAPLE, which is a different function entirely: that function (`sign`) returns the sign of the leading coefficient of the polynomial input to `sign`.

CAUTION: (In General)

This general approach to computing matrix functions can be slow for large exact or symbolic matrices (because manipulation of symbolic representations of the eigenvalues using `RootOf` (typically encountered for $n \geq 5$) can be expensive), and on the other hand can be unstable for floating-point matrices, as is well-known, especially those with nearly multiple eigenvalues. However, for small or for structured matrices this approach can be very useful and can give insight.

12 Matrix stability

As defined in Section §3.9, a matrix is (negative) *stable* if all its eigenvalues are in the left half plane (in this subsection, "stable" means "negative stable"). In MAPLE, one may test this by direct computation of the eigenvalues (if the entries of the matrix are numeric) and this is likely faster and more accurate than any purely rational operation based test such as the Hurwitz criterion. If, however, the matrix contains symbolic entries, then one usually wishes to know for what values of the parameters the matrix is stable. We may obtain conditions on these parameters by using the `Hurwitz` command of the `PolynomialTools` package on the characteristic polynomial.

Examples:

Negative of `gallery(3)` from Matlab.

```
> A := -Matrix( [[-149,-50,-154], [537,180,546], [-27,-9,-25]] ):
> E := Matrix( [[130, -390, 0], [43, -129, 0], [133,-399,0]] ):
> AtE := A - t*E;
```

$$\begin{bmatrix} 149-130\,t & 50+390\,t & 154 \\ -537-43\,t & -180+129\,t & -546 \\ 27-133\,t & 9+399\,t & 25 \end{bmatrix} \tag{9}$$

For which t is that matrix stable?

```
> p := CharacteristicPolynomial( AtE, lambda );
> PolynomialTools[Hurwitz]( p, lambda, 's', 'g' );
```

This command returns "FAIL", meaning that it cannot tell whether p is stable or not; this is only to be expected as t has not yet been specified. However, according to the documentation, all coefficients of λ returned in `s` must be positive, in order for p to be stable. The coefficients returned are

$$\left[\frac{\lambda}{6+t}, 1/4\,\frac{(6+t)^2\,\lambda}{15+433453\,t+123128\,t^2}, \frac{\left(60+1733812\,t+492512\,t^2\right)\lambda}{(6+t)\,(6+1221271\,t)}\right] \tag{10}$$

and analysis (not given here) shows that these are all positive if and only if $t > -6/1221271$.

Acknowledgements

Many people have contributed to linear algebra in Maple, for many years. Dave Hare and David Linder deserve particular credit, especially for the LinearAlgebra package and its connections to CLAPACK, and have also greatly helped our understanding of the best way to use this package. We are grateful to Dave Linder and to Jürgen Gerhard for comments on early drafts of this article.

References

[CGH+96] Robert M. Corless, Gaston H. Gonnet, D. E. G. Hare, David J. Jeffrey, and Donald E. Knuth. On the Lambert W function. *Advances in Computational Mathematics*, 5:329–359, 1996.

[CJ92] R.M. Corless and D.J. Jeffrey. Well... it isn't quite that simple. *SIGSAM Bulletin*, 26(3):2–6, 1992.

[CJ97] R.M. Corless and D.J. Jeffrey. The Turing factorization of a rectangular matrix. *SIGSAM Bulletin*, 31(3):20–28, 1997.

[Cor93] Robert M. Corless. Six, lies, and calculators. *The American Mathematical Monthly*, 100(4):344–350, April 1993.

[Cor04] Robert M. Corless. Generalized companion matrices in the Lagrange basis. In Laureano Gonzalez-Vega and Tomas Recio, editors, *Proceedings EACA*, pages 317–322, June 2004.

[CW04] Robert M. Corless and Stephen M. Watt. Bernstein bases are optimal, but, sometimes, Lagrange bases are better. In *Proceedings of SYNASC, Timisoara*, pages 141–153. MITRON Press, September 2004.

[Goo61] I.J. Good. The colleague matrix, a Chebyshev analogue of the companion matrix. *Quart. J. Math.*, 12:61–68, 1961.

[JV05] Gudbjorn F. Jónsson and Steven Vavasis. Solving polynomials with small leading coefficients. *Siam Journal on Matrix Analysis and Applications*, 26(2):400–414, 2005.

[Win04] Joab R. Winkler. The transformation of the companion matrix resultant between the power and Bernstein polynomial bases. *Appl. Numer. Math.*, 48(1):113–126, 2004.